The purpose of this file is to illustrate a potential issue with fluence to failure.

There are two ways to measure a fluence to failure during SEE testing.

1) stop the test when an event occurs and record the measured fluence as the fluence to failure

2) measure the time of an event and compute the fluence to failure as t_{meas}(total fluence/total run time)

In method 2) the run is stopped at some time after the event of interest. Both methods will yield the same measure for cross section. But, the statistics for cross section are different.

This arises because fluence is a Poisson distributed variable but time to an event will be an exponentially distributed variable. Specifically, if the number of events in time interval t is a Poisson distributed variable with rate parameter λ, then the time to an event will have an exponential distribution with characteristic time $1/\lambda$.

To illustrate first generate a set of random values from a Poisson distribution with rate λ and a set of random values from an exponential distribution with characteristic time $1/\lambda$. The code below generates 1000 values from each distribution with λ set to 100.

```
In[•]:= λ = 100;
τ = RandomReal[ExponentialDistribution[1 / λ], {1000}];
n = RandomInteger[PoissonDistribution[λ], {1000}];
```

We can think of each value in the two sets above as a single run. And we can look at how the average for each set changes as the number of samples increases. Here is a plot for the Poisson distributed values.

In[55]:=
```
Labeled[ListPlot[Accumulate[n] / Range@1000, Frame → True, ImageSize → Large,
  PlotRange → All, FrameLabel → {"Number of samples", "Average"}],
 "Fluence to upset (Poisson)", Top]
```

Out[55]=



And here is the same plot for the exponential values

In[54]:=
```
Labeled[ListPlot[Accumulate[τ] / Range@1000, Frame → True, ImageSize → Large,
  PlotRange → All, FrameLabel → {"Number of samples", "Average"}],
 "Time to upset (exponential)", Top]
```

Out[54]=

Time to upset (exponential)



In both cases, the mean value converges to 100 as expected and both plots have similar appearance. Now look at the trend for average cross section computed as 1 divided by the same values. First the Poisson distributed values

In[57]:=
```
Labeled[ListPlot[Accumulate[1 / n] / Range@1000, Frame → True, ImageSize → Large,
   PlotRange → All, FrameLabel → {"Number of samples", "Average"}],
 "1/Fluence to upset (Poisson)", Top]
```

Out[57]=

1/Fluence to upset (Poisson)



This plot also seems intuitive. The average is converging and converges to 0.01 the inverse of the value chosen for λ. But look what happens for the exponential values

In[56]:=
```
Labeled[ListPlot[Accumulate[1 / τ] / Range@1000, Frame → True, ImageSize → Large,
  PlotRange → All, FrameLabel → {"Number of samples", "Average"}],
 "1/Time to upset (exponential)", Top]
```

Out[56]=

1/Time to upset (exponential)



For the exponential case, there are large jumps in the mean value that occur at random points and the average after 1000 "runs" is a bit more than 0.1 much larger than 0.01 that seems more intuitive. The reason is the distribution for 1/x does not have a mean when x comes from an exponential distribution

The above results can also be obtained from theory as well. First, to show a constant times an exponentially distributed variable has an exponential distribution

In[68]:= $d = \text{TransformedDistribution}\left[\alpha\, x,\ x \approx \text{ExponentialDistribution}\left[\frac{1}{\lambda}\right]\right]$

Out[68]= $\text{ExponentialDistribution}\left[\frac{1}{\alpha\, \lambda}\right]$

The point of this is simply to show fluence to failure has an exponential distribution when computed using method 2) above which simply multiplies the measured time (exponentially distributed variable) by a constant (average flux).

Using the same Mathematica function we find

In[69]:= $f = \text{TransformedDistribution}\left[\frac{1}{x},\ x \approx \text{ExponentialDistribution}\left[\frac{1}{\lambda}\right]\right]$

Out[69]= $\text{FrechetDistribution}\left[1,\ \frac{1}{\lambda}\right]$

So, the distribution for 1/x has a Frechet distribution with parameters 1 and 1/λ when x has an exponential distribution with parameter 1/λ. And

In[71]:= **Mean[*f*]**

Out[71]= ∞

showing the mean of a Frechet distribution with parameters 1 and 1/λ does not have a mean